
1.1 Change 1

	Testcase name
	TC 13.3.2.2

	Protocol
	Multilayer

	Function Name
	f_TC_13_3_2_2_GERAN 

	Reason for change
	During this test case the UE will recieve downlink data on EUTRAN, preform reselection to GERAN, perform RAU procedure and then loop back the data in the uplink. Following RAU, the current TTCN implementation waits for 10s to allow the UE to initiate XID negotiation. If the UE does not initiate XID negotiation but instead attempts to send the pending uplink data before this 10s timer has expired, the test case will fail.

	Summary of change
	Channel request may be returned from f_GERAN_LLC_XID and is passed to f_GERAN_Recieve_UplinkDataStart

	Source of change
	MultiLayer_Procedures_EG_GERAN.ttcn

	MCC160 Comment
	Accepted


Before:
	   function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

  { /* Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var PDP_ContextStatus v_PDP_Context_Status;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    var octetstring v_IPv4IPv6_IcmpEchoReply := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());      /* @sic R5s120470 sic@ */

    //Initialise test case

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);

    v_RAC := f_GERAN_RAC_Get(geran_Cell24);

    v_LAC := f_GERAN_LAC_Get(geran_Cell24);

    //Receieve auth parameters and set cell power level of cell 24 to suitable according to TS 36.508 subclause 6.2.2.1

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell ?

    //f_GERAN_CreateTCH (geran_Cell24);

    //@siclog "Step 2" siclog@

    //@siclog "Step 3-5" siclog@

    //Receive RAU Request Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24,

                                             car_G_LLC_UnitData_IND (cr_G_RA_UpdReqAny (cr_GMM_UpdateType('0'B,

                                                                                                          (tsc_I_UpdateTypeRAOnly, tsc_I_UpdateTypeCombined, tsc_I_UpdateTypeCombinedWithIMSI)), // @sic R5-125545 sic@

                                                                                        cr_RAI (omit),

                                                                                        cr_PTMSI_SignatureAny ifpresent,

                                                                                        ?)));

    //Check PDP context status (shall be indicated as not PDP-INACTIVE)

    v_PDP_Context_Status := v_LLCDataInd.unit.msg.rauReq.pDP_ContextStatus;

    if (match(v_PDP_Context_Status, cs_PDP_ContextStatusInactive)){

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test case 13.3.2.2 Step 3 - PDP Context Status is incorrect/PDP-inactive");

    }

    else {

      f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 3");

    }

    v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc(v_LLCDataInd.unit.msg.rauReq);

    if (v_UpdateType == tsc_I_UpdateTypeCombinedWithIMSI or v_UpdateType == tsc_I_UpdateTypeCombined) {

      v_SecurityInfo := f_GERAN_SecurityInfo_Get();

      v_SecurityInfo.CSinOtherRAT := true;

      f_GERAN_SecurityInfo_Set(v_SecurityInfo);

    }

    //@siclog "Step 6-9" siclog@

    //Perform Authentication

    f_GERAN_GPRS_Authentication (geran_Cell24);

    //@siclog "Step 10-11" siclog@

    //Send RAU Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType),

                                                                                    cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                    cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                                                    v_PDP_Context_Status)));

    //@siclog "Step 12-13" siclog@

    //Receive RAU Complete Message

    f_GERAN_LLC_XID (geran_Cell24, true);
    f_GERAN_SNDCP_Config (1, 1, 5, 11); // @sic R5s12861 sic@

    //@siclog "Step 14" siclog@

    //Void

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    f_GERAN_Receive_UplinkDataStart (geran_Cell24); // @sic R5s120570 sic@

    a_GERAN_Receive_SNDCPData (geran_Cell24, v_IPv4IPv6_IcmpEchoReply); // @sic R5s120570, R5s120861, R5s120919 sic@

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Switch/power off UE and release cell

    f_GERAN_Postamble (geran_Cell24, G1_IDLE);

  }



After:
	    function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

  { /* Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var PDP_ContextStatus v_PDP_Context_Status;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    var template (omit) G_L2_DATAMESSAGE_IND v_ChanReq := omit;
    var octetstring v_IPv4IPv6_IcmpEchoReply := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());      /* @sic R5s120470 sic@ */

    //Initialise test case

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_SNDCP_Config (1, 1, 5, 11); // @sic R5s12861, Hellen sic@

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);

    v_RAC := f_GERAN_RAC_Get(geran_Cell24);

    v_LAC := f_GERAN_LAC_Get(geran_Cell24);

    //Receieve auth parameters and set cell power level of cell 24 to suitable according to TS 36.508 subclause 6.2.2.1

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell ?

    //f_GERAN_CreateTCH (geran_Cell24);

    //@siclog "Step 2" siclog@

    //@siclog "Step 3-5" siclog@

    //Receive RAU Request Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24,

                                             car_G_LLC_UnitData_IND (cr_G_RA_UpdReqAny (cr_GMM_UpdateType('0'B,

                                                                                                          (tsc_I_UpdateTypeRAOnly, tsc_I_UpdateTypeCombined, tsc_I_UpdateTypeCombinedWithIMSI)), // @sic R5-125545 sic@

                                                                                        cr_RAI (omit),

                                                                                        cr_PTMSI_SignatureAny ifpresent,

                                                                                        ?)));

    //Check PDP context status (shall be indicated as not PDP-INACTIVE)

    v_PDP_Context_Status := v_LLCDataInd.unit.msg.rauReq.pDP_ContextStatus;

    if (match(v_PDP_Context_Status, cs_PDP_ContextStatusInactive)){

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test case 13.3.2.2 Step 3 - PDP Context Status is incorrect/PDP-inactive");

    }

    else {

      f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 3");

    }

    v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc(v_LLCDataInd.unit.msg.rauReq);

    if (v_UpdateType == tsc_I_UpdateTypeCombinedWithIMSI or v_UpdateType == tsc_I_UpdateTypeCombined) {

      v_SecurityInfo := f_GERAN_SecurityInfo_Get();

      v_SecurityInfo.CSinOtherRAT := true;

      f_GERAN_SecurityInfo_Set(v_SecurityInfo);

    }

    //@siclog "Step 6-9" siclog@

    //Perform Authentication

    f_GERAN_GPRS_Authentication (geran_Cell24);

    //@siclog "Step 10-11" siclog@

    //Send RAU Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType),

                                                                                    cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                    cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                                                    v_PDP_Context_Status)));

    //@siclog "Step 12-13" siclog@

    //Receive RAU Complete Message

    v_ChanReq := f_GERAN_LLC_XID (geran_Cell24, true);
    //@siclog "Step 14" siclog@

    //Void

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    f_GERAN_Receive_UplinkDataStart (geran_Cell24, v_ChanReq);
    a_GERAN_Receive_SNDCPData (geran_Cell24, v_IPv4IPv6_IcmpEchoReply); // @sic R5s120570, R5s120861, R5s120919 sic@

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Switch/power off UE and release cell

    f_GERAN_Postamble (geran_Cell24, G1_IDLE);

  }


1.2 Change 2

	Testcase name
	TC 13.3.2.2

	Protocol
	Multilayer

	Function Name
	f_GERAN_LLC_XID ()

	Reason for change
	During this test case the UE will recieve downlink data on EUTRAN, preform reselection to GERAN, perform RAU procedure and then loop back the data in the uplink. Following RAU, the current TTCN implementation waits for 10s to allow the UE to initiate XID negotiation. If the UE does not initiate XID negotiation but instead attempts to send the pending uplink data before this 10s timer has expired, the test case will fail.

	Summary of change
	Channel request for two phase acces is handled in f_GERAN_LLC_XID () and retuned from the function if received.

	Source of change
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	Accepted


Before:
	   function f_GERAN_LLC_XID(GERAN_CellId_Type p_CellId,

                           boolean p_RAUComplete,

                           template LLCUnitDataIndType p_Expected_G_RA_UpdComplete := cr_G_RA_UpdComplete) runs on GERAN_PTC

  {

    var B10_Type v_BCCH_ARFCN := f_GERAN_GPRSFreq_Get(p_CellId); // @sic R5s120587 sic@

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_XID;

    var G_RLC_DATAMESSAGE_IND v_G_RLC_ControlMsg_IND;

    var TLLI v_NewTLLI, v_TLLI;

    timer t_Wait10S := 10.0;

    t_Wait10S.start;

    alt {

      [p_RAUComplete]  G_LLC.receive (car_G_LLC_UnitData_IND (p_Expected_G_RA_UpdComplete)) // @sic R5s120837 sic@

        {

          repeat;

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND (p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny)) -> value v_G_RLC_ControlMsg_IND

        {

          v_NewTLLI := v_G_RLC_ControlMsg_IND.ctrl.msg.pktResReq.tlli;

          v_TLLI := f_GERAN_TLLI_Get(p_CellId);

          if (v_NewTLLI != v_TLLI)  //if the TLLI hasn't changed, don't bother calling LLCAssignReq

            {//Assign the value received from the MS

              f_LLC_TLLI_Assign (p_CellId, v_TLLI, v_NewTLLI, dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming data (L3 message)

              f_GERAN_OldTLLI_Set(p_CellId, v_TLLI);

              f_GERAN_TLLI_Set(p_CellId, v_NewTLLI);

            }

          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cds_PacketUplinkAssignment_TwoPhase(v_BCCH_ARFCN,

                                                                                                                                  substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6), // @sic R5s120587 sic@

                                                                                                                                  v_NewTLLI))); // @sic R5s120587 sic@

          repeat;

        }

      [] G_LLC.receive (car_G_LLC_XID_IndAny)

        -> value v_XID

        {

          t_Wait10S.stop;

          // Wait for end of TBF

          // RACE CONDITION: Do we need to check the flag here ???

          G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION

          fl_SendXIDRes (p_CellId, cas_G_LLC_XID_Req(f_GERAN_TLLI_Get(p_CellId), v_XID.xid.sAPI, omit, enum2int(p_CellId))); // @sic R5s130195 Baseline Moving sic@

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          // RACE CONDITION: Do we need to set a flag here ???

          repeat;

        }

      [] G_L2.receive (car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase))

        -> value v_ChanReq

        {

          fl_ULTBFOnePhase (p_CellId, v_ChanReq.access); // @sic R5s120050 MCC160 implementation sic@

          repeat;

        }

      [] t_Wait10S.timeout {} // No XID coming

    }

  }


After:
	function f_GERAN_LLC_XID(GERAN_CellId_Type p_CellId,

                           boolean p_RAUComplete,

                           template LLCUnitDataIndType p_Expected_G_RA_UpdComplete := cr_G_RA_UpdComplete) runs on GERAN_PTC return template (omit) G_L2_DATAMESSAGE_IND
  {

    var B10_Type v_BCCH_ARFCN := f_GERAN_GPRSFreq_Get(p_CellId); // @sic R5s120587 sic@

    var template (omit) G_L2_DATAMESSAGE_IND v_ReturnChanReqTemplate := omit; 
    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_XID;

    timer t_Wait10S := 10.0;

    t_Wait10S.start;

    alt {

      [p_RAUComplete]  G_LLC.receive (car_G_LLC_UnitData_IND (p_Expected_G_RA_UpdComplete)) // @sic R5s120837 sic@

        {

          repeat;

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND (p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        {

          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment(v_BCCH_ARFCN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6)))); // @sic R5s120587 sic@

          repeat;

        }

      [] G_LLC.receive (car_G_LLC_XID_IndAny)

        -> value v_XID

        {

          t_Wait10S.stop;

          // Wait for end of TBF

          // RACE CONDITION: Do we need to check the flag here ???

          G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION

          fl_SendXIDRes (p_CellId, cas_G_LLC_XID_Req(f_GERAN_TLLI_Get(p_CellId), v_XID.xid.sAPI, omit, enum2int(p_CellId))); // @sic R5s130195 Baseline Moving sic@

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          // RACE CONDITION: Do we need to set a flag here ???

          repeat;

        }

      [] G_L2.receive (car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase))

        -> value v_ChanReq

        {

          fl_ULTBFOnePhase (p_CellId, v_ChanReq.access); // @sic R5s120050 MCC160 implementation sic@

          repeat;

        }

      [] G_L2.receive (car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqTwoPhase)) 
        -> value v_ChanReq

        {

          t_Wait10S.stop;

          v_ReturnChanReqTemplate := v_ChanReq;

        }
      [] t_Wait10S.timeout {} // No XID coming

    }

    return v_ReturnChanReqTemplate; 
  }



1.3 Change 3

	Testcase name
	TC 13.3.2.2

	Protocol
	Multilayer

	Function Name
	f_GERAN_Receive_UplinkDataStart ()

	Reason for change
	During this test case the UE will recieve downlink data on EUTRAN, preform reselection to GERAN, perform RAU procedure and then loop back the data in the uplink. Following RAU, the current TTCN implementation waits for 10s to allow the UE to initiate XID negotiation. If the UE does not initiate XID negotiation but instead attempts to send the pending uplink data before this 10s timer has expired, the test case will fail.

	Summary of change
	If a channel request that was previously received during f_GERAN_LLC_XID () it can be passed in to this function.

	Source of change
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	Accepted


Before:
	function f_GERAN_Receive_UplinkDataStart(GERAN_CellId_Type p_CellId) runs on GERAN_PTC
  {

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_CPHY_CONFIG_CNF v_G_CL1_ComingFN_CNF;

    var G_RLC_DATAMESSAGE_IND v_G_RLC_ControlMsg_IND;

    var RFN v_StartingTime;

    var TLLI v_NewTLLI, v_TLLI;

    var B5_Type v_Gamma := f_GERAN_GAMMA_Get(p_CellId);

    var B10_Type v_GPRSFreq := f_GERAN_GPRSFreq_Get(p_CellId);

    var integer v_RR_RA;

    var RFN v_RFN;

      // @sic R5s120521 sic@

      G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, (cr_ChanReqOnePhase, cr_ChanReqTwoPhase) )) -> value v_ChanReq;

    // finish configuring the TBF

    v_RFN := v_ChanReq.access.rfn;

    v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

    if ((111 > v_RR_RA) or (v_RR_RA > 127)) {// establishment cause should be one or two phase access

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");

    }

    v_G_CL1_ComingFN_CNF := f_GERAN_CPHY_CONFIG_ComingFN(cas_G_CL1_ComingFN_REQ( p_CellId, tsc_PhyCh1, tsc_PACCH_F));

    v_StartingTime := v_G_CL1_ComingFN_CNF.comingFN;

    // Send IA

    G_L2.send (cas_G_L2_UNITDATA_REQ (p_CellId,

                                      tsc_PhyCh0,

                                      tsc_AGCH, 15,

                                      cs_G_RFN_Omit,

                                      cas_ImmAss (cs_P_ImmediateAssignment ('0'B,

                                                                            v_GPRSFreq,

                                                                            v_RR_RA,

                                                                            v_RFN,

                                                                            substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6), // @sic R5s120587 sic@

                                                                            cds_IARO_SingleBlkPUA( v_Gamma,

                                                                                         (v_StartingTime.t1_ & v_StartingTime.t3 & v_StartingTime.t2))))));  // @sic GP-110839 sic@

    // Configure the Uplink TBF for Dynamic Allocation, USF 0 on timeslot 4

    f_GERAN_RLC_CONFIG_Common(cas_ActivateTBF_UplinkDyn(p_CellId, 1), dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming car_G_RLC_ControlMsg_IND

    // Now wait for the PRR

    alt {

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        -> value v_G_RLC_ControlMsg_IND

        {

          v_NewTLLI := v_G_RLC_ControlMsg_IND.ctrl.msg.pktResReq.tlli;

          v_TLLI := f_GERAN_TLLI_Get(p_CellId);

          if (v_NewTLLI != v_TLLI)  //if the TLLI hasn't changed, don't bother calling LLCAssignReq

            {//Assign the value received from the MS

              f_LLC_TLLI_Assign (p_CellId, v_TLLI, v_NewTLLI, dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming data (L3 message)

              f_GERAN_OldTLLI_Set(p_CellId, v_TLLI);

              f_GERAN_TLLI_Set(p_CellId, v_NewTLLI);

            }

          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cds_PacketUplinkAssignment_TwoPhase(v_GPRSFreq,

                                                                                                                                  substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6), // @sic R5s120587 sic@

                                                                                                                                  v_NewTLLI)));

        }

     }

  }


After:
	function f_GERAN_Receive_UplinkDataStart(GERAN_CellId_Type p_CellId, template (omit) G_L2_DATAMESSAGE_IND p_ChanReq := omit) runs on GERAN_PTC //
  {

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_CPHY_CONFIG_CNF v_G_CL1_ComingFN_CNF;

    var G_RLC_DATAMESSAGE_IND v_G_RLC_ControlMsg_IND;

    var RFN v_StartingTime;

    var TLLI v_NewTLLI, v_TLLI;

    var B5_Type v_Gamma := f_GERAN_GAMMA_Get(p_CellId);

    var B10_Type v_GPRSFreq := f_GERAN_GPRSFreq_Get(p_CellId);

    var integer v_RR_RA;

    var RFN v_RFN;

    if ( isvalue (p_ChanReq)) {
      v_ChanReq := valueof( p_ChanReq); // need to do this as parameter can be omitted

    } else {
      // @sic R5s120521 sic@

      G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, (cr_ChanReqOnePhase, cr_ChanReqTwoPhase) )) -> value v_ChanReq;

    }
    // finish configuring the TBF

    v_RFN := v_ChanReq.access.rfn;

    v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

    if ((111 > v_RR_RA) or (v_RR_RA > 127)) {// establishment cause should be one or two phase access

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");

    }

    v_G_CL1_ComingFN_CNF := f_GERAN_CPHY_CONFIG_ComingFN(cas_G_CL1_ComingFN_REQ( p_CellId, tsc_PhyCh1, tsc_PACCH_F));

    v_StartingTime := v_G_CL1_ComingFN_CNF.comingFN;

    // Send IA

    G_L2.send (cas_G_L2_UNITDATA_REQ (p_CellId,

                                      tsc_PhyCh0,

                                      tsc_AGCH, 15,

                                      cs_G_RFN_Omit,

                                      cas_ImmAss (cs_P_ImmediateAssignment ('0'B,

                                                                            v_GPRSFreq,

                                                                            v_RR_RA,

                                                                            v_RFN,

                                                                            substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6), // @sic R5s120587 sic@

                                                                            cds_IARO_SingleBlkPUA( v_Gamma,

                                                                                         (v_StartingTime.t1_ & v_StartingTime.t3 & v_StartingTime.t2))))));  // @sic GP-110839 sic@

    // Configure the Uplink TBF for Dynamic Allocation, USF 0 on timeslot 4

    f_GERAN_RLC_CONFIG_Common(cas_ActivateTBF_UplinkDyn(p_CellId, 1), dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming car_G_RLC_ControlMsg_IND

    // Now wait for the PRR

    alt {

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        -> value v_G_RLC_ControlMsg_IND

        {

          v_NewTLLI := v_G_RLC_ControlMsg_IND.ctrl.msg.pktResReq.tlli;

          v_TLLI := f_GERAN_TLLI_Get(p_CellId);

          if (v_NewTLLI != v_TLLI)  //if the TLLI hasn't changed, don't bother calling LLCAssignReq

            {//Assign the value received from the MS

              f_LLC_TLLI_Assign (p_CellId, v_TLLI, v_NewTLLI, dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming data (L3 message)

              f_GERAN_OldTLLI_Set(p_CellId, v_TLLI);

              f_GERAN_TLLI_Set(p_CellId, v_NewTLLI);

            }

          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cds_PacketUplinkAssignment_TwoPhase(v_GPRSFreq,

                                                                                                                                  substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6), // @sic R5s120587 sic@

                                                                                                                                  v_NewTLLI)));

        }

     }

  }

